BIOPAC Systems, Inc.
42 Aero Camino, Goleta, CA 93117
Tel (805) 685-0066 | Fax (805) 685-0067
info@biopac.com | **www.biopac.com**

10.26.2020

Application Note 290: Feedback Control Using BIOPAC Basic Scripting

This application note presents an Exercise in Feedback Control using BBS. Sections include Background, Setup, Algorithm, Results, and Appendices A (channel settings for graph template) and B (code).

1.0 Background

This exercise is intended to show that one can implement feedback control to external devices using MP160/AMI100D and BIOPAC Basic Scripting (BBS). This technique is useful for applications that need/require stability in temperature, flow rate, and pressure.

2.0 Setup

Figure 1 shows the schematic for feedback control. The Flow Measurement signal is processed through CH 8 of the AMI100D then filtered via CH 40 (LPF). The signal from CH 40 is sampled in BBS during OnIdleAcquisition and adjusted as necessary to maintain the desired flowrate. The adjusted signal is routed through OUT0 then to the Flow Source control port (Figure 2).



Figure 1 Schematic of Feedback Control of Flowrate



Figure 2 Flow meter signal fed to CH 8; OUT0 fed to flow source control port.

To correctly applied voltage feedback to the flow source, a mapping of flow rate to voltage is needed. Figure 2 shows that mapping. Also, a regression fit to the mapping points is derived for use in the control macro code.



Figure 3 Mapping of Flow Rate (abscissa) vs Control Voltage (ordinate)

For use in the control code, the regression equation is modified as follows, *where* F = Flow, V = Voltage:

```
;relation between flowrate and voltage set to stimulator lpm --> voltage
;found through regression equation using discrete data points
V = (-0.0000002*F^3) + (0.0002*F^2) + (0.000006*F) - 0.0115
```

3.0 Algorithm

The algorithm is as follows:

OnIdleAcquisition

- during an idle portion of the acquisition, selected the entire graph then extract the mean value of the last second of data

```
K = Ticks / 60      ;get the time of sampling
t = K – T           ;difference between sampling and base time
if t > 2            ;if > 2 seconds, determine adjustment
Select Wave 40
Edit SelectAll
Set HCursor HCursor2 - 1, HCursor2
X = Mean Measured Flow Rate
```

- compare to desired flow rate and apply adjustments to the voltage is necessary

```
;find the ratio of measured to desired flow rate
R = X / F
V = V / R ;adjust the voltage to maintain desired flow
MP100 Set Out0 V

T = K   ;reset the base time
else
;Continue
endif
```

End

4.0 Test Results

Figures 4a, 4b show an acquisition where the Flow Rate, specified at 87 LPM, is maintained when leaks are introduced into the system.

After initial settling, a bleed valve is opened to allow leakage of the gas and to cause a drop in the flow rate. The control algorithm rightly corrects for this by adjusting the flow source voltage to produce MORE flow to compensate for the loss.

When the bleed valve is closed, the flow meter senses a transitory increase in flow to which the algorithm compensates by adjusting the flow source voltage to produce LESS flow.



Figure 4a Flow Rate adjustment via feedback control



Figure 4b Flow Rate adjustment via feedback control. Expanded view

Appenidx A | Channel Settings for Graph Template

Acquisition Settings

MP160

Serial #: MP160 001F54

Graph: Feedback Flow Ref.acq

Sampling rate: 1000 samples/sec

Record and replace previous data; save to memory.

Acquisition length: 2 hours (7200001 samples)

Smart Amplifier Configuration:

Acquired channels:

- Analog Channel 8
  - Label: Flow Reference
  - Scaling: 0.00030518 volts -> 0 lpm, 10.0003 volts -> 300 lpm
  - Units: lpm
- Calculation Channel 0
  - Label: Flow 10Hz LPF - LT
  - Units: lpm
  - Filter - IIR
  - Source channel: Analog Channel 8
  - Low pass at 10 Hz
  - Q: 0.707

Appendix B | Code

```
Lessons
OnOpenFile

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;Ventilator feedback control.bbs;;;;;;;;;;;;;;;;;;;;
F = 25  ;lpm
T = 0
CloseAllGraphs

Open "", r, W$
if r <> 1
   Prompt "File could not be opened!", "OK"
else
   Select Window W$
   Call "EnterFlow"
endif
End
EnterFlow
;set the flow rate
GetNumber "Enter the desired flow rate: ", "lpm", F, e, false
if e <> 0
  Call "ExitProgram"
else
  Call "SetSTIM"
endif
End
SetSTIM
;relation between flowrate and voltage set to stimulator lpm --> voltage
;found through regression equation using discrete data points
V = (-0.0000002*F^3) + (0.0002*F^2) + (0.000006*F) - 0.0115

Call "NewButtons"
End
StartMPAcq
RemoveAllButtons
CreateButton "Done" "Done"
MP100 StartAcq
End
```

```
OnStartAcquisition
MP100 Set Out0 V
T = Ticks / 60
End
Done
MP100 StopAcq
MP100 Set Out0 0
Call "NewButtons"
End
NewButtons
RemoveAllButtons
CreateButton "Enter Flowrate" "EnterFlow"
CreateButton "Start Acq" "StartMPAcq"
CreateButton "Exit Program" "ExitProgram"
End
ExitProgram
;;AppExit
Select Window W$
CloseGraph
End
OnIdleAcquisition
K = Ticks / 60     ;get the time of sampling
t = K - T          ;difference between sampling and base time
if t > 2           ;if > 2 seconds, determine adjustment
   Select Wave 40
   Edit SelectAll
   Set HCursor HCursor2 - 1, HCursor2
   X = Mean

   ;find the ratio of measured to desired flow rate
   R = X / F
   V = V / R             ;adjust the voltage to maintain desired flow
   MP100 Set Out0 V

   T = K     ;reset the base time
else
   ;Continue
endif
End
```